IDL: Manipulating Scientific Data Format Files and Data Visualization

Jules Kouatchou and Bruce Van Aartsen

Jules.Kouatchou@nasa.gov, bruce.vanaartsen@nasa.gov

Goddard Space Flight Center
ASTG - Code 606

June 20, 2017

Training Objectives

We want to introduce:

- Basic concepts of scientific data format files
- Manipulation of netCDF and HDF files
- Data Visualization

Create a File

For each file format, we will show examples on how to:

- Define dimensions
- Define variables
- Add variable and global attributes
- Write data

Read a File

For each file format, we will show examples on how to:

- Read a variable
- Read attributes
- Read a subset of a variable
- Read a file a list all the variables, their dimension, type, etc.

Plotting Data

We will show examples on how to read data from (netCDF, HDF) files and the do following types of plots:

- Contours
- Zonal mean heights
- Various time series

Materials

The presentation slides are in the *Slides/* directory and the scripts for:

Beginners are in the directories:

```
BasicSyntax/      If_Loops/
Array/            Structures/
Proc_Function/    File_IO/       Plots/
```

This training are in the directories:

```
netCDF/   HDF4/    HDF5/
```

Data Files

In case you downloaded the "LARGE" file containing the data files, make sure that you move:

- The two netCDF files into the directory: *netCDF/ncFiles/*
- The two HDF4 files into the directory: *HDF4/hdFiles/*

Journaling

```
IDL> journal, 'my_journal.pro'
IDL> x = -1.0 + 0.1*FINDGEN(21)
IDL> y = exp(x)
IDL> plot, x, y
IDL> journal
```

```
IDL> @my_journal.pro
```

Scientific Data Format

Advantages of using a standard scientific data format:

- Portability
- Self-describing data (*metadata*)
- Mixed data types
- Widely available read/write software

When creating a scientific data format file, we need to make an extra effort to include enough metadata so that we can understand the file after it is created.

Standard Variables Attributes

Attribute Name	Definition
long_name	A text string that describes a variable in detail.
units	A text string that describes the units of a variable.
valid_range	A two-element array containing the valid minimum and maximum values for a variable (e.g., [-10.0, 250.0]).
scale_factor	A multiplier to be applied to a variable after it is read
add_offset	An offset to be added to a variable after it is read, and after sca1e_factor (if present) is applied.
_FillValue	A value indicating that no data were written.

Manipulating netCDF Files

Overview of netCDF

- The network Common Data Form (netCDF) was developed by the Unidata Program Center at the University Corporation for Atmospheric Research in Boulder, Colorado.
- The basic building blocks of netCDF files are variables, attributes, and dimensions:
  1. Variables are scalars or multidimensional arrays.
  2. Attributes contain supplementary information about a single variable (variable attribute) or an entire file (global attribute).
  3. Dimensions are *long* scalars that record the size of one or more variables.

Sample Structure of a netCDF File

```
Name:
       image.nc
       Dimensions.
       xsize = 1200
       ysize = 600
Variables:
       byte image[xsize, ysize]
              long_name = 'Imager visible channel'
              units = 'Counts'
              valid_range = 0, 255
       double time[ysize]
              long_name = 'Seconds since 0000 UTC, Jan 1 1970'
              units = 'seconds'
              valid_range = 0.0D+0, 10.0D+308
Global Attributes:
       title = 'GOES Image'
       history =  'Created Wed Jul 14 14.15.01 1993'
```

Routines for Writing netCDF Files

NCDF_CREATE - Create a new file that is put into define mode.

NCDF_DIMDEF - Create dimensions for the file

NCDF_VARDEF - Define the variables to be used in the file.

NCDF_ATTPUT - Optionally, use attribute to describe the data

NCDF_CONTROL, /ENDEF - Call NCDF_CONTROL and set the ENDEF keyword to leave define mode and enter data mode.

NCDF_VARPUT - Write the appropriate data to the netCDF file.

NCDF_CLOSE - Close the file.

Writing a Simple file

```
ny    =  12
nx    =   6
data = LINDGEN(nx, ny)

ncfid    = NCDF_CREATE('simple_xy.nc', /CLOBBER)
xid   = NCDF_DIMDEF(ncfid, 'x', nx)
yid   = NCDF_DIMDEF(ncfid, 'y', ny)
vid   = NCDF_VARDEF(ncfid, 'data', [xid, yid], /LONG)
NCDF_CONTROL, ncfid, /ENDEF

NCDF_VARPUT, ncfid, 'data', data

NCDF_CLOSE, ncfid
```

Routines for Reading netCDF Files

NCDF_OPEN - Create an existing netCDF file..

NCDF_INQUIRE - Find the format of the netCDF file.

NCDF_DIMINQ - Retrieve the names and sizes of dimensions in the file.

NCDF_VARINQ - Retrieve the names, types, and sizes of variables in the file.

NCDF_ATTNAME - Optionally, retrieve attribute names..

NCDF_ATTINQ - Optionally, retrieve the types and lengths of attributes.

NCDF_ATTGET - Optionally, retrieve the attributes.

NCDF_VARGET - Read the data from the variables.

NCDF_CLOSE - Close the file.

Reading a Simple file

```

ncfid = NCDF_OPEN('simple_xy.nc')    ; Open file

NCDF_VARGET, ncfid, 'data', data     ; Read variable 'data

NCDF_CLOSE, ncfid                    ; Close file
```

Reading an Attribute

```
varid = NCDF_VARID(ncfid, variable_name)
NCDF_ATTGET, ncfid, varid, var_attribute_name, var_attri
print, var_attribute_value

NCDF_ATTGET, ncfid, glob_attribute_name, glob_attibute_va
print, glob_attibute_value
```

Reading a Subset of a Variable

```
varid = NCDF_VARID(ncfid, variable_name)
NCDF_VARGET, ncfid, varid, data, $
        OFFSET = [...], $
        COUNT  = [...], $
        STRIDE  = [...]
```

OFFSET: The first element in each dimension to be read (zero-based; default is [0,0,...,0])

COUNT: The number of elements to be read in each dimension (default is from the current OFFSET to the last element in each dimension)

STRIDE: The sampling interval along each dimension (default is [1,1,...,1], which samples every element)

Exercise on netCDF

Modify the file *NC_printVariablesInfo.pro* to:

- Not list the dimension variables from the list of variables
- List the data type of each variable (before dimensions)

Manipulating HDF4 Files

Overview of HDF4

- The Hierarchical Data Format (HDF) is a data file format designed by the National Center for Supercomputing Applications (NCSA).
- Some features of HDF:
  1. Supports a variety of data types.
  2. Makes it possible for programs to obtain information about the data from the data file itself, rather than from another source.
  3. Standardizes the format and descriptions of many types of commonly used data sets, such as raster images and scientific data..

Commonly used HDF4 Scientific Data Set Routines

Name	Purpose
hdf_sd_start()	Open a HDF file in SDS mode
hdf_sd_end	Close a HDF file in SDS mode
hdf_sd_nametoindex()	Return variable index
hdf_sd_select()	Return variable identifier
hdf_sd_getdata	Read variable data
hdf_sd_endaccess	End access to a variable
hdf_sd_attrfind()	Return variable/global attribute index
hdf_sd_attrinfo	Read variable/global attribute data
hdf_sd_fileinfo	Return file information
hdf_sd_getinfo	Return variable information
hdf_sd_create()	Create a variable
hdf_sd_dimgetid()	Create a dimension
hdf_sd_dimset	Set dimension information
hdf_sd_adddata	Write variable data
hdf_sd_attrset	Write attribute data

Writing a Simple HDF4 File

```
ny    =  12
nx    =   6
data = LINDGEN(nx, ny)

hdfid    = HDF_SD_START('simple_xy.hdf', /CREATE)

vid   = HDF_SD_CREATE(hdfid, 'data', [nx, ny], /LONG)

xid   = HDF_SD_DIMGETID(vid, 0)
HDF_SD_DIMSET, xid, NAME = 'x'

yid   = HDF_SD_DIMGETID(vid, 1)
HDF_SD_DIMSET, yid, NAME = 'y'

HDF_SD_ADDDATA, vid, data

HDF_SD_ENDACCESS, vid
HDF_SD_END, hdfid
```

Reading a Simple HDF4 File

```
hdfid = HDF_SD_START('simple_xy.hdf')

; Find the index of the sds to read using its name
index = HDF_SD_NAMETOINDEX(hdfid, 'data')

; Select it
vid = HDF_SD_SELECT(hdfid, index)

; Get data set information including
; dimension information
HDF_SD_GetInfo, vid, name = 'data', natts = num_attribut
                                    ndim = num_dims, di

; Read the data
HDF_SD_GETDATA, vid, data

HDF_SD_END, hdfid
```

Reading a Subset of a Variable

```
index = HDF_SD_NAMETOINDEX(hdfid, variable_name)
vid = HDF_SD_SELECT(hdfid, index)
HDF_SD_GETDATA, vid, data, $
        START  = [...], $
        COUNT  = [...], $
        STRIDE = [...]
 HDF_SD_ENDACCESS, vid
```

START: The first element in each dimension to be read (zero-based; default is [0,0,...,0])

COUNT: The number of elements to be read in each dimension (default is from the current START to the last element in each dimension)

STRIDE: The sampling interval along each dimension (default is [1,1,...,1], which samples every element)

Reading an Attribute from a HDF4 File

```
index = HDF_SD_NAMETOINDEX(hdfid, variable_name)
vid = HDF_SD_SELECT(hdfid, index)
attindex = HDF_SD_ATTRFIND(vid, attribute_name)
HDF_SD_ATTRINFO, vid, attindex, data=attribute_value
HDF_SD_ENDACCESS, vid
print, attribute_value
```

Exercise on HDF4

Modify the file *HDF4_printVariablesInfo.pro* to:

- Not list the dimension variables from the list of variables
- List the data type of each variable (before dimensions)

Manipulating HDF5 Files

Overview of HDF5

- The Hierarchical Data Format (HDF) is a data file format designed by the National Center for Supercomputing Applications (NCSA).
- Some features of HDF5:
  1. Can represent very complex data objects and a wide variety of metadata.
  2. Portable file format.
  3. A rich set of integrated performance features that allow for access time and storage space optimizations.
  4. Tools and applications for managing, manipulating, viewing, and analyzing the data in the collection.

HDF5 Data Model

- **Groups** - provide structure among objects
- **Datasets** - where the primary data goes
  - Data arrays
  - Rich set of datatype options
  - Flexible, efficient storage and I/O
- **Attributes** - for metadata

Everything else is built essentially from these parts.

Routines for HDF5 Interfaces

Interface	**Routine Class**	**Example**
Attributes	H5A	H5A_CREATE, H5A_OPEN
		H5A_CLOSE, H5A_READ, H5A_WRITE
Datasets	H5D	H5D_READ
File	H5F	H5F_OPEN
Group	H5G	H5G_CREATE, H5G_OPEN, H5G_CLOSE
Reference	H5R	
Dataspace	H5S	H5S_CLOSE
Datatype	H5T	H5T_CREATE, H5T_CLOSE

Basic IDL HDF5 Functions

```
H5F_CREATE (H5F_OPEN)          ; create (open) File

  H5S_CREATE                   ; create dataSpace

     H5D_CREATE (H5D_OPEN)     ; create (open) Dataset

        H5D_READ, H5D_WRITE    ; access Dataset

     H5D_CLOSE                 ; close Dataset

  H5S_CLOSE                    ; close dataSpace

H5F_CLOSE                      ; close File
```

Writing a Simple HDF5 File

```
h5fid    = H5F_CREATE('simple_xy.h5') ; Create the file

; Get data type and space needed to create the dataset
datatype_id  = H5T_IDL_CREATE(data)
dataspace_id = H5S_CREATE_SIMPLE(size(data,/DIMENSIONS))

; create dataset in the output file
dataset_id = H5D_CREATE(h5fid, 'data', datatype_id, $
                        dataspace_id)

; write data to dataset
H5D_WRITE,dataset_id,data

; close all open identifiers
H5D_CLOSE,dataset_id
H5S_CLOSE,dataspace_id
H5T_CLOSE,datatype_id
H5F_CLOSE,h5fid
```

Reading a Simple HDF5 File

```
h5fid = H5F_OPEN('simple_xy.h5')

; Open the data set
vid = H5D_OPEN(h5fid, 'data')

; Read the data
data = H5D_READ(vid)
help, data

; Close all open identifiers
H5D_CLOSE, vid
H5F_CLOSE, h5fid
```

Reading a Subset of a Variable

```
h5fid = H5F_OPEN(file_name)
vid   = H5D_OPEN(h5fid, 'data')
vspace = H5D_GET_SPACE(vid)
H5S_SELECT_HYPERSLAB, vspace, START=[...], $
        COUNT=[...], BLOCK=[...], STRIDE=[...], /RESET
 rspace = H5S_CREATE_SIMPLE(count)
 data = H5D_READ(vid, FILE_SPACE=vspace, $
                 MEMORY_SPACE=rspace)
```

START: The first element in each dimension to be read (zero-based; default is [0,0,...,0])

COUNT: The number of elements to be read in each dimension (default is from the current START to the last element in each dimension)

BLOCK: The size of each block, typically a single element at the time.

STRIDE: The sampling interval along each dimension (default is [1,1,...,1], which samples every element)

Reading a Subset of a Variable - II

Assume that we have a 3D array *arr* and we want to extract the entries:

```
arr[3:3:1, 5:49:2, 0:49:3]
```

We then have:

```
START  = [3, 5, 0]
COUNT  = [1, 23, 17]
STRIDE = [1, 2, 3]
BLOCK  = [1, 1, 1]
```

Reading an Attribute from a HDF5 File

```
index = HDF_SD_NAMETOINDEX(hdfid, variable_name)
vid = HDF_SD_SELECT(hdfid, index)
attindex = HDF_SD_ATTRFIND(vid, attribute_name)
HDF_SD_ATTRINFO, vid, attindex, data=attribute_value
HDF_SD_ENDACCESS, vid
print,  attribute_value
```

IDL Utility for HDF5 Tables

http://www.github.com/superchromix/wmb_lib

Plotting

Simple Contour Plot

```
CONTOUR, data, lons, lats
```

See file *netCDF/plt_contour_temp.pro*.

Simple Contour Plot with Basic Levels

```
num_levels = 6
CONTOUR, data, lons, lats, $
         NLevels=num_levels, $
         C_Labels=Replicate(1, num_levels)
```

See file *netCDF/plt_contour_temp_levels1.pro*.

Simple Contour Plot with Levels (from Data)

```
num_levels = 6
step = (MAX(data) - MIN(data) / num_levels
mylevels = IndGen(num_levels) * step + Min(data)
CONTOUR, data, lons, lats, $
         Levels=mylevels, $
         C_Labels=Replicate(1, num_levels)
```

See file *netCDF/plt_contour_temp_levels2.pro*.

Filled Contour Plot

```
num_levels = 6
ncolors = num_levels + 1
step = (Max(data) - Min(data)) / num_levels
mylevels = IndGen(num_levels) * step + Min(data)
c_levels = mylevels
c_colors = indgen(ncolors) + bottom
loadct, 33, ncolors=ncolors, bottom=1

CONTOUR, data, lons, lats, $
         levels=c_levels, c_colors=c_colors, /fill
CONTOUR, data, lons, lats, $
         Levels=mylevels, $
         C_Labels=Replicate(1, num_levels), /overplot
```

See file *netCDF/plt_filled_contour_temp.pro*.

Add Colorbar

```
CONTOUR, data, lons, lats, $
         levels=c_levels, c_colors=c_colors, /fill $
         Position=[0.125, 0.20, 0.95, 0.7]

colorbar, N_levels = num_levels, Colors = c_colors, $
          Labels = c_levels, Levels=c_levels, $
          Position=[0.125, 0.92, 0.95, 0.96]
```

See file *netCDF/plt_colobar_contour_temp.pro*.

Cylindral Map Projection

```
Map_Set, /Cylindrical,  /hires, color = 0, $
      Position=[0.1, 0.1, 0.9, 0.8], $
      Limit=[Min(lats), Min(lons), Max(lats), Max(lons)],
      /ADVANCE, /isotropic, /GRID, /noborder

CONTOUR, data, lons, lats, levels=c_levels, $
          c_colors=c_colors, /Cell_fill, /Overplot, $

CONTOUR, data, lons, lats, /Overplot, $
          color = 0, levels=c_levels, c_labels = c_labels

Map_Grid, Color=1, GLINESTYLE=2
MAP_CONTINENTS,/COASTS,color=0,MLINETHICK=2
```

See file *netCDF/plt_cyl_map_contour_temp.pro*.

Polar Map Projection

```
Map_Set, /Stereographic, center_lat, center_lon, $
          color = 0, Position=[0.1, 0.1, 0.9, 0.8], $
          /Advance, /Continents, /Grid, /Isotropic, $
          /NoErase, /Horizon, /NoBorder

CONTOUR, data, lons, lats, levels=c_levels, $
          c_colors=c_colors, /Cell_fill, /Overplot

CONTOUR, data, lons, lats, /Overplot, $
          color = 0, levels=c_levels

Map_Grid, Color=1, GLINESTYLE=2
MAP_CONTINENTS, /COASTS, color=0, MLINETHICK=2
```

See file *netCDF/plt_pol_map_contour_temp.pro*.

Plotting Subdomain

```
min_lon = -20.0   & max_lon = 120.0
min_lat = -30.0   & max_lat = 40

imin_lat = Value_Locate(lats, min_lat)
imax_lat = Value_Locate(lats, max_lat)
imin_lon = Value_Locate(lons, min_lon)
imax_lon = Value_Locate(lons, max_lon)
lons=lons[imin_lon:imax_lon] & lats=lats[imin_lat:imax_l
nlons=N_ELEMENTS(lons) & nlats=N_ELEMENTS(lats)

NCDF_VARGET, ncfid, vName, var, $
    OFFSET=[imin_lon, imin_lat, ref_level, ref_time], $
    COUNT=[nlons, nlats, 1, 1], STRIDE=[1, 1, 1, 1]
```

See file *netCDF/plt_subdomain_uwind.pro*.

Plotting Zonal Mean Height

```
    NCDF_VARGET, ncfid, vName, var, $
                 OFFSET = [0, 0, 0, time_rec], $
                 COUNT  = [nlons, nlats, nlevs, 1], $
                 STRIDE = [1, 1, 1, 1]
var = MEAN(var, DIMENSION=1)

CONTOUR, var, lats, levs
```

See file *netCDF/plt_zonal_mean_height_uwind.pro*.

Wind Vector Time Series

See file *netCDF/plt_wind_vector_series.pro*

Streamlines Time Series

See file *netCDF/plt_streamlines_series.pro*

References I

Kenneth P. Bowman.
An Introduction to Programming with IDL: Interactive Data Language.
Elsevier Inc, 2005.

David W. Fanning.
Traditional IDL Graphics.
Coyote Book Publishing, 2011.

Lilian Gumley.
Practical IDL Programming.
Morgan Kaufmann, 2001.